

Analysis and Measurement of Intrinsic Noise in Op Amp Circuits Part IV: Introduction to SPICE Noise Analysis

by Art Kay, Senior Applications Engineer, Texas Instruments Incorporated

In Part III of this series, we did a hand analysis on a simple operational amplifier (op amp) circuit. In this, Part IV, we use a circuit simulation package called TINA SPICE to analyze op amp circuits. (A free version of TINA SPICE: TINA-TI can be downloaded on the Texas Instruments web site by typing in TINA under search at <http://www.ti.com>). TINA SPICE can do the traditional types of simulation associated with SPICE packages (eg, dc, transient, frequency domain analysis, noise analysis and more). Furthermore, TINA-TI comes loaded with many TI analog macromodels.

In Part IV, we introduce TINA noise analysis and show how to prove that the op amp macromodel accurately models noise. It is important to understand that some models may not properly model noise. To check this a simple test procedure is used. This problem is solved by developing our own models using discrete noise sources and a generic op amp.

Test The Op Amp Model Noise Accuracy

Fig. 4.1 shows the test circuit used to verify the accuracy of the noise model of an op amp.




Fig. 4.1: Configure Noise Test Circuit (Set Up CCV1 Gain)

CCV1 is a current-controlled voltage source that we will use to convert the noise current to noise voltage. The reason for this conversion is that the *output noise analysis* in TINA strictly looks at noise voltage. The gain of the CCV1 must be set to "1" as shown so that current will be translated directly into voltage. The op amp is put into a voltage follower configuration so that the input noise will be reflected to the output. The two output measurement nodes *voltage_noise* and *current_noise* are identified by TINA as nodes

used to generate noise plots. The source VG1 is added because TINA requires an input source for noise analysis. I configure this source as sinusoidal, but this is not critical for noise analysis (see Fig. 4.2).




Fig. 4.2: Configure Noise Test Circuit (Set Up Signal Source)

Next, a noise analysis must be done. Select *Analysis\Noise Analysis* from the pull-down menu as shown in Fig. 4.3.




Fig. 4.3: Run the "Noise Analysis" Option

This brings up the Noise Analysis form. Enter the start and end frequency of interest. This frequency range is determined by the specifications of the op amp under test. For this example, examining the specifications for the OPA227, shows that a frequency range of 0.1 Hz to 10 kHz is appropriate because this is the range that noise is specified over. Next, select the *Output Noise* option under *Diagrams*. The *Output Noise* option generates a spectral density curve for each measurement node (meter) in the circuit. Hence, when this analysis is run, we will get two spectral density plots: one for the *Voltage Noise* node and one for the *Current Noise* node.

Results from the noise analysis are shown in Fig. 4.4. A few simple tricks can be used to convert these curves into a more useful format. First, use *Separate Curves* under the *View* menu. Next, click on the Y-axis and select the *Logarithmic* scale. Set the lower limit and the upper limit to the appropriate range (round to powers of 10). The number of ticks should be adjusted to $1 + \text{Number_of_Decades}$. In this case, we have three decades (ie 100f to 100p). Therefore, we need four ticks (see Fig. 4.5).




Fig. 4.4: Simple Tricks Clean Up Format (Separate Curves)




Fig. 4.5: Simple Tricks Clean Up Format (Change to "Log Scale")

The simulation results are compared to the OPA227 datasheet in Fig. 4.6. Note that the two are virtually identical. Thus, the TINA-TI model for the OPA227 accurately models noise. The same procedure was followed for the OPA627 model. Fig. 4.7 shows the results for this test. The OPA627 model does not pass the test. The current noise spectral density of the model is approximately $3.5E-21A/\sqrt{Hz}$ and the specification is $2.5E-15A/\sqrt{Hz}$. Furthermore, the voltage noise in the model shows no $1/f$ region. In the next section, we will build a model for this op amp that properly models noise.




Fig. 4.6: OPA227 Passes the Model Test

Fails the Test

PARAMETER	OPA627BM, BP, SM OPA637BM, BP, SM			UNITS
	MIN	TYP	MAX	
NOISE				
Input Voltage Noise				
Noise Density, $f = 10\text{Hz}$		15	40	nV/√Hz
$f = 100\text{Hz}$		8	20	nV/√Hz
$f = 1\text{kHz}$		5.2	8	nV/√Hz
$f = 10\text{kHz}$		4.5	6	nV/√Hz
Voltage Noise, BW = 0.1Hz to 10Hz		0.6	1.6	μVp-p
Input Bias Current Noise				
Noise Density, $f = 100\text{Hz}$		1.6	2.5	fA/√Hz
Current Noise, BW = 0.1Hz to 10Hz		30	40	fAp-p





Fig. 4.7: OPA627 Fails the Model Test

Build Your Own Noise Model

In Part II, the op amp noise model was introduced consisting of an op amp, a voltage noise source, and a current noise source. We will build this noise model using discrete noise sources and a generic op amp. The discrete noise sources were developed for Texas Instruments by Bill Sands (*Analog and Rf models*), and can be downloaded for your use from the TI website at <http://www.ti.com>. Search for "TINA-TI Application Schematics" and look for the "Noise Analysis" folder. The TINA Macro listing is also given in Appendix 4.1 and 4.2.

Fig. 4.8 illustrates the circuit used to create the noise model. Note that it is in the test mode configuration that we used previously and has one current noise source connected between the inputs. Strictly speaking, there are two current noise sources. The degree, however, to which these sources are correlated is not always clear from the product data sheet. Furthermore, the magnitude of these sources is different in current-current-feedback amplifiers. These topics will be covered in greater detail in a future article. We will customize this circuit to properly model the noise characteristics of the OPA627.




Fig. 4.8: Op Amp Noise Model Using Discrete Noise Sources

First, we must configure the noise voltage source. This is done by right-clicking on the source and selecting *Enter Macro* (see Fig. 4.9). Enter *Macro* to bring up a text editor that has the SPICE macromodel listing for the source. Fig. 4.10 shows the ".PARAM" information that needs to be edited to match the datasheet. Note that NLF is the noise spectral density magnitude (in $\text{nV}/\sqrt{\text{Hz}}$) of a point in the $1/f$ region. FLW is the frequency of the selected point.

Enter magnitude of 1/f and broadband noise into the macro.





Fig. 4.9: Enter The Macro For The Noise Voltage Source

1/f Region

Look for a point in the 1/f region. Enter the frequency and magnitude at this point



```

opa627 noise source mod:U1 [MACRO] - Schematic Editor
File Edit Insert View Analysis T&M Tools Help
Basic/Switches/Meters/Sources/Semiconductors/Optoelectronics/S...
* BEGIN PROG NSE NANOVOLT/RT-HZ
.SUBCKT VNSE 30 40
* BEGIN SETUP OF NOISE GEN - NANOVOLT/RT-
* INPUT THREE VARIABLES
* SET UP VNSE 1/F
* NV/RHZ AT 1/F FREQ
.PARAM NLF=50
* FREQ FOR 1/F VAL
.PARAM FLW=1
* SET UP VNSE FB
* NV/RHZ FLATBAND
.PARAM NVR=5
* END USER INPUT
* START CALC VALS
.PARAM GLF={FLW^0.25*NLF/1164}
.PARAM RNV={1.184*NVR^2}
.MODEL DVN D KF={FLW^0.5/1E11} IS=1.0E-16
  
```

Fig. 4.10: Enter The Data For The 1/f Region

Next, we need to enter the broadband noise spectral density. This is done using the NVR parameter. Note that a frequency is not required because the magnitude of the broadband noise is constant over all frequencies (see Fig. 4.11). After entering the noise information, we must compile and close the SPICE text editor. Press the check box and note that a “Successfully compiled.” message will appear in the status bar. Under the “File” menu, select “Close” to return to the schematic editor (see Fig. 4.12).




Fig. 4.11: Enter the Data for the Broadband Region

The same procedure must be performed on the noise current source. For this example, there is no 1/f noise for the current source. In this case, the broadband and 1/f ".PARAM" are set to the same level (2.5 fA/√Hz). The 1/f frequency is set to some very low frequency out of the range of normal interest such as 0.001 Hz (see Fig. 4.13).

Compile the Macro

After macro is compiled press "file > close" and return to schematic editor.




Fig. 4.12: Compile "Macro" and "Close"

Follow the same procedure for current noise. This example has no 1/f component (set FLWF = 0.001).

```

* BEGIN PROG NSE_FEMTO AMP/RT-HZ
.SUBCKT FEMTO 1 2
* BEGIN SETUP OF NOISE GEN - FEMPTOAMP:
* INPUT THREE VARIABLES
* SET UP INSE 1/F
* FA/RHZ AT 1/F FREQ
.PARAM NLFF=2.5
* FREQ FOR 1/F VAL
.PARAM FLWF=0.001
* SET UP INSE FB
* FA/RHZ FLATBAND
.PARAM NVRF=2.5
* END USER INPUT
* START CALC VALS
    
```

PARAMETER	OPA627BM, BP, SM OPA637BM, BP, SM			UNITS
	MIN	TYP	MAX	
NOISE				
Input Voltage Noise				
Noise Density, f = 10Hz		15	40	nV/√Hz
f = 100Hz		8	20	nV/√Hz
f = 1kHz		5.2	8	nV/√Hz
f = 10kHz		4.5	6	nV/√Hz
Voltage Noise, BW = 0.1Hz to 10Hz		0.6	1.6	μVp-p
Input Bias Current Noise				
Noise Density, f = 100Hz		1.6	2.5	fA/√Hz
Current Noise, BW = 0.1Hz to 10Hz		30	60	fAp-p

Fig. 4.13: Enter Data for Current Noise Source

Now that both noise sources have been properly configured, we must edit some ac parameters in the generic op amp model. Specifically, the open-loop gain and dominant pole must be entered. These must be entered because they can affect the closed-loop bandwidth of the amplifier which in turn affects the noise performance of the circuit. The open-loop gain is normally given in dB in the datasheet. Equation 4.1 is used to convert from dB to linear gain. Equation 4.2 is used to calculate the dominant pole in the Aol curve. Example 4.1 does the dominant calculation for the OPA627. The dominant pole is shown graphically in Fig. 4.14.

$$OLG = 10^{\left(\frac{Ndb}{20}\right)}$$

Where

OLG = the Open Loop Gain in V/V

Ndb = the Open Loop Gain in dB

Eq. 4.11

$$\text{Dominant_Pole} = \frac{GBW}{OLG}$$

Where

Dominant_Pole = the first pole in the op amp Open Loop Gain curve

GBW = The Gain Bandwidth Product

OLG = the Open Loop Gain in V/V

Eq. 4.2

$$OLG = 10^{\frac{Ndb}{20}} = 10^{\left(\frac{120}{20}\right)} = 1 \cdot 10^6 \text{ V/V}$$

OPA627 Data Sheet

$$\text{Dominant_Pole} = \frac{GBW}{OLG} = \frac{16\text{MHz}}{1 \cdot 10^6} = 16\text{Hz}$$

				dB
OPEN-LOOP GAIN				
Open-Loop Voltage Gain	112	120		dB
Over Specified Temperature	106	117		dB
SM Grade	100	114		dB
FREQUENCY RESPONSE				
Slew Rate: OPA627	40	55		V/μs
OPA637	100	135		V/μs
Settling Time: OPA627 0.01%		550		ns
0.1%		450		ns
OPA637 0.01%		450		ns
0.1%		300		ns
Gain-Bandwidth Product: OPA627		16		MHz
OPA637		80		MHz
Total Harmonic Distortion + Noise		0.00003		%

Example 4.1: Find Linear Open Loop Gain And Dominant Pole For OPA627




Fig. 4.14: Dominant Pole On Gain Vs Frequency Plot

Next, the generic op amp model must be edited to include the open-loop gain and dominant pole. This is done by double clicking on the op amp symbol and pressing the *Type* button. This brings up the *Catalog Editor*. From the *Catalog Editor*, change the *Open loop gain* to match what we calculated using the results from Example 4.1. Fig. 4.15 summarizes this procedure.

The figure shows a circuit diagram of an operational amplifier (OP1) with a feedback network consisting of resistors R1 (1k) and R2 (100k). The op amp is powered by a 15V supply (V1) and has a 15V output (V2). A voltage source VG1 is connected to the non-inverting input, and a voltage source U1 is connected to the inverting input. The output is labeled V627. Two dialog boxes are overlaid on the circuit. The top dialog box is titled "OP1 - Operational amplifier" and shows the "Type" field set to "IOPAMP". The bottom dialog box is titled "Catalog Editor" and shows the "Model Parameters" for the "IOPAMP" model. The parameters are: Open loop gain [-] 1M, Input resistance [Ohm] 2M, Output resistance [Ohm] 55, Maximum slew rate [V/s] 5M, Dominant pole [Hz] 10, Second pole [Hz] 100G, Input offset voltage [V] 0, Input bias current [A] 0, Input offset current [A] 0, Offset voltage tco [V/C] 0, Current doubling int. [C] 10, and Outp. offs. lim. [Vcc+][V] 2. A red box contains the following steps:

1. Double Click on Op-Amp
2. Press "Type" Button
3. Edit "Open loop gain" and "Dominant Pole" according to Op-Amp data sheet

Fig. 4.15: Edit The Generic Op Amp

Now the op amp noise model is complete. Fig. 4.16 shows the result of running the test procedure on the model. As expected, the new model matches the datasheet exactly.




Fig. 4.16: The New Hand-Built Model Passes The Model Test

Use TINA To Analyze The Circuit From Part III

Fig. 4.17 illustrates the schematic for OPA627 entered into Tina SPICE. Note that the noise sources and op amp were developed in Part IV. Also note that the resistors R_f (R_2) and R_1 match the example circuit from Part III.




Fig. 4.17: OPA627 Example Circuit

A Tina SPICE noise analysis is run by selecting *Analysis\Noise Analysis* from the pull down menu. This brings up the noise analysis form. Select the *Output Noise* and *Total Noise* options on the noise analysis form. The *Output Noise* option will generate a noise spectral density plot for all measurement nodes (ie nodes with meters). *Total Noise* will generate a plot of the integrated power spectral density curve. The total noise curve allows us to determine the rms output noise voltage for this circuit. Fig. 4.18 shows how to run the noise analysis.




Fig. 4.18: Run The Noise Analysis

The results of the TINA noise analysis are shown in Figs. 4.19 and 4.20. Fig. 4.19 shows the noise spectral density at the output of the amplifier (ie Output Noise). This curve combines all the noise sources and includes the effects of noise gain, and noise bandwidth. Fig. 4.20 shows the total noise at the output of the amplifier for a given bandwidth. This curve was derived by integrating the power spectral density curve (ie the voltage spectral density squared). Note that the curve is at a constant $323 \mu\text{V}_{\text{rms}}$ at high frequency. This result compares very well to the rms noise computed in Part III (ie the computed noise was $324 \mu\text{V}$). Note that the noise is a constant value because of the op amp bandwidth limitations.




Fig. 4.19: Result For Output Noise Plot




Fig. 4.20: Result For Total Noise Plot

Summary and Preview

In this TechNote we introduced a circuit simulation package named TINA SPICE. We developed a simple test procedure using TINA that can be used to check an op amp model. In some cases, models will fail this procedure, which is why we developed our own model using discrete noise sources and a generic op amp. We also used TINA to calculate noise for the example circuit that we created in our hand analysis in Part III. In Part V, we will investigate methods for measuring noise. In particular, we will physically measure the noise computed in the previous sections.

Acknowledgments

Special thanks to all of the technical insights from the following individuals:

- Rod Burt, Senior Analog IC Design Manager, TI
- Bruce Trump, Manager Linear Products, TI
- Tim Green, Applications Engineering Manager, TI
- Neil Albaugh, Senior Applications Engineer, TI
- Bill Sands, Consultant, *Analog and Rf Models*:
<http://www.home.earthlink.net/%7ewksands/>

References

- 1.) Robert V Hogg, and Elliot A Tanis, *Probability and Statistical Inference*, 3rd Edition, Macmillan Publishing Company
- 2.) C D Motchenbacher, and J A Connelly, *Low-Noise Electronic System Design*, a Wiley-Interscience Publication

About The Author

Arthur Kay is a senior applications engineer at Texas Instruments where he specializes in the support of sensor signal conditioning devices. Prior to TI, he was a semiconductor test engineer for Burr-Brown and Northrop Grumman Corp. He graduated from Georgia Institute of Technology with an MSEE in 1993. Art can be reached at ti_artkay@list.ti.com

Appendix 4.1: Voltage Noise Macro

```
* BEGIN PROG NSE NANO VOLT/RT-HZ
.SUBCKT VNSE 1 2
* BEGIN SETUP OF NOISE GEN - NANOVOLT/RT-HZ
* INPUT THREE VARIABLES
* SET UP VNSE 1/F
* NV/RHZ AT 1/F FREQ
.PARAM NLF=15
* FREQ FOR 1/F VAL
.PARAM FLW=10
* SET UP VNSE FB
* NV/RHZ FLATBAND
.PARAM NVR=4.5
* END USER INPUT
* START CALC VALS
.PARAM GLF={PWR(FLW,0.25)*NLF/1164}
.PARAM RNV={1.184*PWR(NVR,2)}
.MODEL DVN D KF={PWR(FLW,0.5)/1E11} IS=1.0E-16
* END CALC VALS
I1 0 7 10E-3
I2 0 8 10E-3
D1 7 0 DVN
D2 8 0 DVN
E1 3 6 7 8 {GLF}
R1 3 0 1E9
R2 3 0 1E9
R3 3 6 1E9
E2 6 4 5 0 10
R4 5 0 {RNV}
R5 5 0 {RNV}
R6 3 4 1E9
R7 4 0 1E9
E3 1 2 3 4 1
C1 1 0 1E-15
C2 2 0 1E-15
C3 1 2 1E-15
.ENDS
• END PROG NSE NANOV/RT-HZ
```


Appendix 4.2: Current Noise Macro

```
* BEGIN PROG NSE FEMTO AMP/RT-HZ
.SUBCKT FEMT 1 2
* BEGIN SETUP OF NOISE GEN - FEMPTOAMPS/RT-HZ
* INPUT THREE VARIABLES
* SET UP INSE 1/F
* FA/RHZ AT 1/F FREQ
.PARAM NLFF=2.5
* FREQ FOR 1/F VAL
.PARAM FLWF=0.001
* SET UP INSE FB
* FA/RHZ FLATBAND
.PARAM NVRF=2.5
* END USER INPUT
* START CALC VALS
.PARAM GLFF={PWR(FLWF,0.25)*NLFF/1164}
.PARAM RNVF={1.184*PWR(NVRF,2)}
.MODEL DVNF D KF={PWR(FLWF,0.5)/1E11} IS=1.0E-16
* END CALC VALS
I1 0 7 10E-3
I2 0 8 10E-3
D1 7 0 DVNF
D2 8 0 DVNF
E1 3 6 7 8 {GLFF}
R1 3 0 1E9
R2 3 0 1E9
R3 3 6 1E9
E2 6 4 5 0 10
R4 5 0 {RNVF}
R5 5 0 {RNVF}
R6 3 4 1E9
R7 4 0 1E9
G1 1 2 3 4 1E-6
C1 1 0 1E-15
C2 2 0 1E-15
C3 1 2 1E-15
.ENDS
* END PROG NSE FEMTO AMP/RT-HZ
```

as published in...

analog **ZONE**